

Язык запросов 1С

Тема 1. Выборка данных из одной таблицы: основы


Оглавление

1. Консоль запросов	1
1.1. Подготовка консоли запросов	1
1.2. Использование консоли запросов	4
2. Основы языка запросов	6
2.1. ВЫБРАТЬ	6
2.2. ИЗ	10
2.3. Псевдонимы	10
2.4. ГДЕ	11
2.4.1. Операторы сравнения: >, <, >=, <=, <>, =	11
2.4.2. Логические операторы: И, ИЛИ, НЕ	12
2.4.3. Оператор МЕЖДУ	13
2.4.4. Арифметические операции: +, -, *, /, %	13
2.4.5. Оператор В	14
2.4.6. Оператор ПОДОБНО	14
2.5. Параметры	15
2.6. Функции работы со строками	16
2.7. Функции работы с датами	17
2.8. УПОРЯДОЧИТЬ ПО	18
2.9. ВЫБОР ... КОГДА ... ТОГДА	18
2.10. Конструктор запросов	20


1. Консоль запросов

1.1. Подготовка консоли запросов


1. Создаем новую конфигурацию «Язык запросов» (часть шагов пропущено)



2. Заходим в Конфигуратор созданной конфигурации




3. Из материалов курса скачиваем файл «Учебная база SQL.dt» и в открывшейся конфигурации «Язык запросов»



4. Открываем файл «Учебная база SQL.dt»




5. Подтверждаем загрузку




6. Перезапускаем конфигуратор



7. Открываем конфигурацию. Наличие нескольких справочников, документов, регистров накопления, регистра сведений и обработки означает, что загрузка выполнена успешно




8. Нажимаем «Начать отладку» (горячая клавиша F5), запускается конфигурация в режиме «1С:Предприятие», и на главной странице отображается Консоль запросов версии 8.3.2.21




1.2. Использование консоли запросов


1. Основные элементы панели списка запросов: 1 – дерево запросов, в каждой ветке которого размещается отдельный запрос, 2 – добавление новой ветки в дерево запросов, 3 – сохранение всего дерева запросов (всех запросов) в текстовый файл, 4 – открытие текстового файла с сохраненным в нем деревом запросов и загрузка содержимого в текущую консоль запросов




2. Основные элементы панели текста запроса: 1 – поле для ввода текста запроса (сами запросы будут рассмотрены далее), 2 – кнопка «Выполнить», запускающая запрос на выполнение, 3 – кнопка «Проверка» используется для синтаксической проверки правильности написания текста запроса и обновления списка параметров запроса, 4 – список основных команд, доступных в тексте запроса.




3. Основные элементы панели параметров запроса: 1 – список всех параметров с установленным значением параметров, 2 – настройка типа параметра



4. Панель результатов запроса: 1 – поле для вывода результатов запроса, 2 – редактирование результатов запроса, 3 – страницы для сохранения результатов разных запросов



- Панель пакетов запросов – используется в случае когда, в панели запросов написано несколько запросов, и необходимо выбрать какие из этих запросов выполнять, а какие нет (на начальном этапе изучения темы не рекомендуется трогать эту панель)



2. Основы языка запросов

2.1. ВЫБРАТЬ

Оператор ВЫБРАТЬ используется для указания какие реквизиты необходимо отобразить.


ПРИМЕР:

ВЫБРАТЬ

Справочник.Страны.Наименование

Запрос вернет названия всех стран, записанных в справочнике «Страны»

РЕЗУЛЬТАТ ЗАПРОСА:



Если необходимо отобразить несколько реквизитов, то они перечисляются через запятую

ПРИМЕР:

ВЫБРАТЬ

Справочник.Страны.Код,

Справочник.Страны.Наименование

Запрос вернет код и названия всех стран, записанных в справочнике «Страны»

РЕЗУЛЬТАТ ЗАПРОСА:

Результат1 (0,001 мс., строк: 9)

Код	Наименование
000000001	Таиланд
000000002	Абхазия
000000003	Греция
000000004	Египет
000000005	Россия
000000006	Мальдивы
000000007	Грузия
000000008	Белоруссия
000000009	Вьетнам

Аналогичным образом можно выбирать реквизиты не только Справочников, но и всех остальных элементов конфигурации, где могут храниться данные – документы, регистры сведений и регистры накопления.

ПРИМЕР:

ВЫБРАТЬ

Документ.Заказы.Договор,
Документ.Заказы.Дата,
Документ.Заказы.Покупатель

Запрос вернет наименование договора, дату его заключения и наименование покупателя, записанные в документе «Заказы»

РЕЗУЛЬТАТ ЗАПРОСА:

Результат1 (0,007 мс., строк: 11)

Договор	Дата	Покупатель
Договор № 15	13.06.2021 23:21:20	ООО "ВЕГА"
Договор №09	13.06.2021 23:21:23	ООО "ОРИОН"
Договор 1/2 от 2021	13.06.2021 23:20:54	ООО "ПЕРСПЕКТИВА"
Договор №17	02.02.2020 23:21:11	ООО "МАГНИТ"
Договор №22	30.06.2021 23:59:00	ООО "ВИТА"
Договор №09	06.07.2021 12:00:00	ООО "ОРИОН"
Договор №22	12.06.2021 9:22:58	ООО "ВИТА"
Договор 1/1 от 2021	05.05.2021 12:00:01	ООО "ПЕРСПЕКТИВА"
Основной договор	12.06.2021 12:00:02	ООО "ЭРА"
Основной договор	05.01.2022 15:41:11	ООО "ЭРА"
Основной договор	10.01.2024 13:06:17	ООО "ЛИДЕР"

ПРИМЕР:

ВЫБРАТЬ

РегистрСведений.Цены.Период,
РегистрСведений.Цены.Товар,
РегистрСведений.Цены.Цена

Запрос вернет дату, когда установлена цена на товар, наименование товара и цена на товар, записанные в регистре сведений «Цены»

РЕЗУЛЬТАТ ЗАПРОСА:

Результат1 (0,004 мс., строк: 15)

Период	Товар	Цена
01.01.2020 0:00:00	Папайя	290
01.01.2021 0:00:00	Папайя	340
01.01.2021 0:00:00	Вино "Радеда"	4 600
01.01.2021 0:00:00	Масло оливковое "Grand di Oliva"	660
01.01.2021 0:00:00	Бедуинский платок	2 200
01.01.2021 0:00:00	Самовар	6 500
01.01.2021 0:00:00	Раа	20 900
01.01.2021 0:00:00	Минеральная вода "Borjomi"	140
10.01.2024 0:00:00	Вино "Vaziani"	4 500
10.01.2024 0:00:00	Минанкари	1 200
10.01.2024 0:00:00	Лапти	400
10.01.2024 0:00:00	Сало	160
10.01.2024 0:00:00	Бедуинский платок	3 800
10.01.2024 0:00:00	Масло оливковое "Grand di Oliva"	900
10.01.2024 0:00:00	Минеральная вода "Borjomi"	145

В операторе ВЫБРАТЬ можно указывать константные значения строк, чисел, дат, арифметические операции, конкатенацию строк и некоторые другие операции

ПРИМЕР:

ВЫБРАТЬ

РегистрНакопления.ИсторияЗаказов.Товар,
РегистрНакопления.ИсторияЗаказов.Количество,
РегистрНакопления.ИсторияЗаказов.Цена,
"Сумма = ",
РегистрНакопления.ИсторияЗаказов.Количество *
РегистрНакопления.ИсторияЗаказов.Цена

Запрос вернет название, количество и цену заказанного товара, записанные в регистре накопления «ИсторияЗаказов». В четвертом столбце везде будет отображаться строка "Сумма = ", а в пятом столбце будет рассчитана стоимость (по формуле количество * цена).

РЕЗУЛЬТАТ ЗАПРОСА:

Результат1 (0,003 мс., строк: 28)

Товар	Количество	Цена	Поле1	Поле2
Масло оливковое "Grand di Oliva"	2	900	Сумма =	1 800
Папайя	45	290	Сумма =	13 050
Вино "Радеда"	2	4 600	Сумма =	9 200
Самовар	5	6 500	Сумма =	32 500
Папайя	1	340	Сумма =	340
Самовар	4	6 500	Сумма =	26 000
Вино "Радеда"	1	4 600	Сумма =	4 600
Папайя	2	340	Сумма =	680
Вино "Радеда"	2	4 600	Сумма =	9 200
Самовар	3	6 500	Сумма =	19 500
Папайя	10	340	Сумма =	3 400
Папайя	2	340	Сумма =	680
Самовар	3	6 500	Сумма =	19 500
Вино "Радеда"	5	4 600	Сумма =	23 000
Папайя	4	340	Сумма =	1 360
Самовар	2	6 500	Сумма =	13 000
Бедуинский платок	2	2 200	Сумма =	4 400
Раа	2	20 900	Сумма =	41 800
Папайя	1	340	Сумма =	340
Бедуинский платок	2	2 200	Сумма =	4 400
Вино "Радеда"	1	4 600	Сумма =	4 600
Самовар	4	6 500	Сумма =	26 000
Папайя	2	340	Сумма =	680
Папайя	2	340	Сумма =	680
Бедуинский платок	3	2 200	Сумма =	6 600
Бедуинский платок	1	3 800	Сумма =	3 800
Бедуинский платок	2	2 200	Сумма =	4 400
Раа	2	20 900	Сумма =	41 800

Иногда в указанном для выборки столбце ячейки не содержат каких-либо данных (пустые ячейки), тогда в результатах запроса выведется значение <NULL>. Если вместо этого стандартного значения требуется вывести какое-то сообщение о том, что поле не заполнено, применяют функцию ЕСТЬNULL, в первом аргументе которого указывается ресурс (столбец таблицы), а во втором аргументе – то сообщение, которое нужно вставить вместо пустой ячейки.

ПРИМЕР:

ВЫБРАТЬ

Справочник.Покупатели.Регион.Наименование,
ЕСТЬNULL(Справочник.Покупатели.Регион.Наименование, "Данные не заполнены")

Запрос вернет два столбца с названиями регионов, однако в первом столбце на месте пустых ячеек будет отображаться <NULL>, а во втором столбце – строка "Данные не заполнены"

РЕЗУЛЬТАТ ЗАПРОСА:

Результат1 (0,009 мс., строк: 25)	
РегионНаименование	Поле1
г. Москва	г. Москва
Республика Адыгея (Адыгея)	Республика Адыгея (Адыгея)
<NULL>	Данные не заполнены
Волгоградская область	Волгоградская область
Брянская область	Брянская область
Томская область	Томская область
<NULL>	Данные не заполнены
г. Москва	г. Москва
г. Санкт-Петербург	г. Санкт-Петербург
Краснодарский край	Краснодарский край
г. Москва	г. Москва
Хабаровский край	Хабаровский край
г. Москва	г. Москва
Магаданская область	Магаданская область
Томская область	Томская область
Московская область	Московская область
г. Москва	г. Москва
г. Санкт-Петербург	г. Санкт-Петербург
г. Санкт-Петербург	г. Санкт-Петербург
г. Москва	г. Москва
Самарская область	Самарская область
Приморский край	Приморский край
г. Москва	г. Москва
Приморский край	Приморский край
г. Москва	г. Москва

Рядом за оператором ВЫБРАТЬ могут быть записаны операторы:

РАЗЛИЧНЫЕ — удаление из результата дублирующие строки.

ПЕРВЫЕ n — вывод первых n записей.

ДЛЯ ИЗМЕНЕНИЯ — блокирует записи таблицы на время транзакции (не актуально для управляемых блокировок). Подробнее в статье ДЛЯ ИЗМЕНЕНИЯ в запросах 1С.

РАЗРЕШЕННЫЕ — выбирает только записи из таблицы, на которые есть права у текущего пользователя.

ПРИМЕР:

ВЫБРАТЬ РАЗЛИЧНЫЕ

Документ.Заказы.Покупатель

Запрос вернет названия покупателей, записанные в документе «Заказы», без повторения одного и того же названия

РЕЗУЛЬТАТ ЗАПРОСА:

Результат1 (0,005 мс., строк: 7)	
Покупатель	
ООО "ВЕГА"	
ООО "ОРИОН"	
ООО "ПЕРСПЕКТИВА"	
ООО "МАГНИТ"	
ООО "ВИТА"	
ООО "ЭРА"	
ООО "ЛИДЕР"	

Оператор ВЫБРАТЬ редко когда используется в одиночку. Обычно его дополняет оператор ИЗ (см. ниже). При использовании в запросе оператора ИЗ после оператора ВЫБРАТЬ можно поставить символ звездочки (умножения), что будет означать все столбцы из указанных в операторе ИЗ данных.

ПРИМЕР:

ВЫБРАТЬ

*

ИЗ

Документ.ВводОстатков

Запрос вернет все реквизиты (в т.ч. стандартные) из документа «ВводОстатков»

РЕЗУЛЬТАТ ЗАПРОСА:

Результат1 (0,019 мс., строк: 2)					
Ссылка	ВерсияДанных	Помет	Номер	Дата	Прове
Ввод остатков 000000001 от 09.01.2020 12:00:00	AAAAAQAAAA	Нет	000000001	09.01.2020 12:00:00	Да
Ввод остатков 000000002 от 10.01.2024 11:39:40	AAAAAQAAAAE	Нет	000000002	10.01.2024 11:39:40	Да

2.2. ИЗ

Оператор ИЗ используется для указания из каких элементов конфигурации необходимо вывести данные

ПРИМЕР:

ВЫБРАТЬ

Справочник.Регионы.Код,
Справочник.Регионы.Наименование

ИЗ

Справочник.Регионы

Запрос вернет реквизиты Код и Наименование из справочника «Регионы»

РЕЗУЛЬТАТ ЗАПРОСА:

Результат1 (0,001 мс., строк: 90)	
Код	Наименование
000000001	Республика Адыгея (Адыгея)
000000002	Республика Башкортостан
000000003	Республика Бурятия
000000004	Республика Алтай
000000005	Республика Дагестан
000000006	Республика Ингушетия
000000007	Кабардино-Балкарская Республика

В операторе ИЗ можно указывать более, чем один элемент конфигурации, но эта ситуация будет рассмотрена позже в теме про соединение таблиц. В настоящей теме во всех запросах в операторе ИЗ будет указана только одна таблица.

2.3. Псевдонимы

Псевдонимы используются для того, чтобы название таблицы или столбца таблицы в запросе записывать более компактно. Псевдонимы задаются оператором КАК, который возможно указать как после названия таблицы в блоке ИЗ, так и после названия столбца, указанного в блоке ВЫБРАТЬ. В первом случае псевдоним вводится для компактного обращения к таблице (в теле самого запроса), во втором случае – для компактного обращения к столбцу таблицы и для задания имени столбца в отображаемой таблице.

ПРИМЕР:

ВЫБРАТЬ

Регион.Код КАК Номер,
Регион.Наименование КАК НазваниеРегиона

ИЗ

Справочник.Регионы КАК Регион

Запрос аналогичен предыдущему запросу, но в заголовках выводимой таблицы отображаются названия «Номер» и «НазваниеРегиона». При этом: 1) за счет псевдонима «Регион» в тексте запроса в операторе ВЫБРАТЬ вместо полного названия таблицы «Справочник.Регионы» указывается просто «Регион», что несколько сокращает и

облегчает понимание текст запроса; 2) псевдонимы «Номер» и «НазваниеРегиона» задают соответствующие (более точные с точки зрения содержания) названия столбцам таблицы при выводе.

РЕЗУЛЬТАТ ЗАПРОСА:



Результат1 (0,013 мс., строк: 90)

Номер	НазваниеРегиона
000000001	Республика Адыгея (Адыгея)
000000002	Республика Башкортостан
000000003	Республика Бурятия
000000004	Республика Алтай
000000005	Республика Дагестан
000000006	Республика Ингушетия
000000007	Кабардино-Балкарская Республика
000000008	Республика Калмыкия

2.4.ГДЕ

Оператор ГДЕ используется для фильтрации строк результатов запроса. Выделяют несколько основных условий для фильтрации строк:

2.4.1. Операторы сравнения: >, <, >=, <=, <>, =

Операторы сравнения в основном используются для сравнения чисел, дат и строк.

ПРИМЕР:

ВЫБРАТЬ

*

ИЗ

РегистрСведений.Цены КАК Цены

ГДЕ

Цены.Цена > 2500

Запрос выводит дату, когда установлена цена на товар, наименование товара и значение цены на товар, если это значение больше 2500. Все остальные данные из регистра сведений «Цены», не удовлетворяющие условию «Цены.Цена > 2500» в результирующую выборку не попадают.

РЕЗУЛЬТАТ ЗАПРОСА:



Результат1 (0,001 мс., строк: 5)

Период	Товар	Цена
01.01.2021 0:00:00	Вино "Радеда"	4 600
01.01.2021 0:00:00	Самовар	6 500
01.01.2021 0:00:00	Раа	20 900
10.01.2024 0:00:00	Вино "Vaziani"	4 500
10.01.2024 0:00:00	Бедуинский платок	3 800

ПРИМЕР:

ВЫБРАТЬ

*

ИЗ

РегистрСведений.Цены КАК Цены

ГДЕ

Цены.Период >= ДАТАВРЕМЯ(2024,01,01,0,0,0)

Запрос выводит дату, когда установлена цена на товар, наименование товара и значение цены на товар, для всех тех случаев, когда значение цены на товар было установлено с 01 января 2024 г. или позже. Примечание: для указания даты в текущем запросе используется функция ДАТАВРЕМЯ, использование одинарных кавычек для задания даты недопустимо.

РЕЗУЛЬТАТ ЗАПРОСА:

Результат1 (0,001 мс., строк: 7)

Период	Товар	Цена
10.01.2024 0:00:00	Вино "Vaziani"	4 500
10.01.2024 0:00:00	Лапти	400
10.01.2024 0:00:00	Минанкари	1 200
10.01.2024 0:00:00	Сало	160
10.01.2024 0:00:00	Масло оливковое "Grand di Oliva"	900
10.01.2024 0:00:00	Бедуинский платок	3 800
10.01.2024 0:00:00	Минеральная вода "Borjomi"	145

2.4.2. Логические операторы: И, ИЛИ, НЕ

Логические (булевы) операторы И и ИЛИ используются для объединения нескольких условий в одно более сложное.

ПРИМЕР:

ВЫБРАТЬ

*

ИЗ

РегистрСведений.Цены КАК Цены

ГДЕ

Цены.Цена >= 4500 И Цены.Цена <= 5000

Запрос выводит дату, когда установлена цена на товар, наименование товара и значение цены на товар, если это значение больше или равно 4500 и меньше или равно 5000 включительно (находится в диапазоне от 4500 до 5000 включительно).

РЕЗУЛЬТАТ ЗАПРОСА:

Результат1 (0,002 мс., строк: 2)

Период	Товар	Цена
01.01.2021 0:00:00	Вино "Радеда"	4 600
10.01.2024 0:00:00	Вино "Vaziani"	4 500

Оператор НЕ используется для задания противоположного (обратного) условия

ПРИМЕР:

ВЫБРАТЬ

*

ИЗ

РегистрСведений.Цены КАК Цены

ГДЕ

НЕ Цены.Период >= ДАТАВРЕМЯ(2021,01,01,0,0,0)

Запрос выводит дату, когда установлена цена на товар, наименование товара и значение цены на товар, для всех тех случаев, когда значение цены на товар было установлено ранее 01 января 2024 г. (условие противоположно условию из 2го примера п.4.1).

РЕЗУЛЬТАТ ЗАПРОСА:

Результат1 (0,002 мс., строк: 8)

Период	Товар	Цена
01.01.2020 0:00:00	Папайя	290
01.01.2021 0:00:00	Папайя	340
01.01.2021 0:00:00	Вино "Радеда"	4 600
01.01.2021 0:00:00	Масло оливковое "Grand di Oliva"	660
01.01.2021 0:00:00	Бедуинский платок	2 200
01.01.2021 0:00:00	Самовар	6 500
01.01.2021 0:00:00	Раа	20 900
01.01.2021 0:00:00	Минеральная вода "Borjomi"	140

2.4.3. Оператор МЕЖДУ

Операторы МЕЖДУ используется для задания условия вхождения значения выбранных данных в интервал между двумя указанными граничными значениями. Границы интервала в отборе учитываются (интервал нестрогий).

ПРИМЕР:

ВЫБРАТЬ
*
ИЗ
РегистрСведений.Цены КАК Цены
ГДЕ
Цены.Цена МЕЖДУ 4500 И 5000

Запрос выводит дату, когда установлена цена на товар, наименование товара и значение цены на товар, если это значение больше или равно 4500 и меньше или равно 5000 включительно (находится в диапазоне от 4500 до 5000 включительно). Запрос аналогичен запросу из 1го примера п.4.2, но с более компактной записью условия.

РЕЗУЛЬТАТ ЗАПРОСА:

Период	Товар	Цена
01.01.2021 0:00:00	Вино "Радеда"	4 600
10.01.2024 0:00:00	Вино "Vaziani"	4 500

2.4.4. Арифметические операции: +, -, *, /, %

Арифметические операции могут использоваться в разных операторах (блоках) запроса, в т.ч. в блоках ВЫБРАТЬ и ГДЕ

ПРИМЕР:

ВЫБРАТЬ
Цены.Период КАК ДатаУстановленияЦены,
Цены.Товар КАК Товар,
Цены.Цена КАК Цена,
Цена * 0.2 КАК НДСНаЕдиницуТовара
ИЗ
РегистрСведений.Цены КАК Цены
ГДЕ
Цена > 2 * 100 И
(Товар.Наименование = "Масло оливковое ""Grand di Oliva"" ИЛИ
Товар.Наименование = "Минеральная вода ""Borjomi"" ИЛИ
Товар.Наименование = "Минанкари")

Запрос выводит дату, когда установлена цена на товар, наименование товара, значение цены на товар и сумму НДС на единицу товара, рассчитанную по ставке 20%. Данные выводятся только для трех товаров, у которых цена больше 200 (т.е. 2 * 100) и если наименование товара имеет одно из трех значений: Масло оливковое "Grand di Oliva", Минеральная вода "Borjomi", Минанкари

РЕЗУЛЬТАТ ЗАПРОСА:

ДатаУстановленияЦены	Товар	Цена	НДСНаЕдиницуТовара
01.01.2021 0:00:00	Масло оливковое "Grand di Oliva"	660	132
10.01.2024 0:00:00	Масло оливковое "Grand di Oliva"	900	180
10.01.2024 0:00:00	Минанкари	1 200	240

2.4.5. Оператор В

Оператор В используется для сравнения сразу с несколькими значениями. Оператор срабатывает, если реквизит равен хотя бы одному из указанных в операторе В значению.

ПРИМЕР:

ВЫБРАТЬ

Цены.Период КАК ДатаУстановленияЦены,
Цены.Товар КАК Товар,
Цены.Цена КАК Цена,
Цена * 0.2 КАК НДСНаЕдиницуТовара

ИЗ

РегистрСведений.Цены КАК Цены

ГДЕ

Цена > 2*100 И
Товар.Наименование В("Масло оливковое ""Grand di Oliva""",
"Минеральная вода ""Borjomi""", "Минанкари")

Запрос выводит дату, когда установлена цена на товар, наименование товара, значение цены на товар и сумму НДС на единицу товара, рассчитанную по ставке 20%. Данные выводятся только для трех товаров, у которых цена больше 200 (т.е. 2 * 100) и если наименование товара имеет одно из трех значений: Масло оливковое "Grand di Oliva", Минеральная вода "Borjomi", Минанкари. Запрос аналогичен запросу из примера п.4.4, но с более компактной записью условия.

РЕЗУЛЬТАТ ЗАПРОСА:



ДатаУстановленияЦены	Товар	Цена	НДСНаЕдиницуТовара
01.01.2021 0:00:00	Масло оливковое "Grand di Oliva"	660	132
10.01.2024 0:00:00	Масло оливковое "Grand di Oliva"	900	180
10.01.2024 0:00:00	Минанкари	1 200	240

2.4.6. Оператор ПОДОБНО

Оператор ПОДОБНО используется в случае, если точное (полное) наименование реквизита неизвестно или требуется отобрать данные по некоторой маске. В операторе ПОДОБНО используются следующие операции:

% - означает произвольное количество символов,

_ - означает один (произвольный) символ,

[] – означает любой символ, перечисленный внутри скобок,

[^] – означает любой символ, кроме тех, что перечислены внутри скобок.

ПРИМЕР:

ВЫБРАТЬ

Цены.Период КАК ДатаУстановленияЦены,
Цены.Товар КАК Товар,
Цены.Цена КАК Цена,
Цена * 0.2 КАК НДСНаЕдиницуТовара

ИЗ

РегистрСведений.Цены КАК Цены

ГДЕ

Цена > 2*100 И
Товар.Наименование ПОДОБНО "М%"

Запрос выводит дату, когда установлена цена на товар, наименование товара, значение цены на товар и сумму НДС на единицу товара, рассчитанную по ставке 20%. Данные выводятся только для трех товаров, у которых цена больше 200 (т.е. 2 * 100) и если

наименование товара начинается с буквы «М» (после буквы М может идти любое количество любых символов). Запрос возвращает тот же самый результат, что в запросах из примеров п.4.4 и п.4.5., однако указанные выше запросы работают только с 3мя наименованиями товаров, а данный запрос – со всем, которые начинаются с буквы «М», соответственно, если в систему будут добавлены новые товары, у которых цена будет больше 200, а наименование начинаться с буквы «М», то эти товары будут текущим запросом также выведены на экран.

РЕЗУЛЬТАТ ЗАПРОСА:

Результат1 (0,002 мс., строк: 3)

ДатаУстановленияЦены	Товар	Цена	НДСНаЕдиницуТовара
01.01.2021 0:00:00	Масло оливковое "Grand di Oliva"	660	132
10.01.2024 0:00:00	Масло оливковое "Grand di Oliva"	900	180
10.01.2024 0:00:00	Минанкари	1 200	240

ПРИМЕР:

ВЫБРАТЬ

*

ИЗ

РегистрСведений.Цены КАК Цены

ГДЕ

Товар.Наименование ПОДОБНО "%[й,",""]%"

Запрос выводит дату, когда установлена цена на товар, наименование товара и значение цены на товар. Данные выводятся только для трех товаров, у которых в наименовании встречается символ двойной кавычки или буква й.

РЕЗУЛЬТАТ ЗАПРОСА:

Результат1 (0,001 мс., строк: 10)

Период	Товар	Цена
01.01.2020 0:00:00	Папайя	290
01.01.2021 0:00:00	Папайя	340
01.01.2021 0:00:00	Вино "Радеда"	4 600
01.01.2021 0:00:00	Масло оливковое "Grand di Oliva"	660
10.01.2024 0:00:00	Масло оливковое "Grand di Oliva"	900
01.01.2021 0:00:00	Бедуинский платок	2 200
10.01.2024 0:00:00	Бедуинский платок	3 800
01.01.2021 0:00:00	Минеральная вода "Borjomi"	140
10.01.2024 0:00:00	Минеральная вода "Borjomi"	145
10.01.2024 0:00:00	Вино "Vaziani"	4 500

2.5. Параметры

Необходимые значения реквизитов можно задавать в самом запросе, а можно с помощью параметров (параметры часто сравнивают с переменными языка 1С). Использование параметров более рекомендуемый способ указания значений.

ПРИМЕР:

ВЫБРАТЬ
*
ИЗ
РегистрСведений.Цены КАК Цены
ГДЕ
Цена.Период >= <ДатаУстановленияЦены

Название: ДатаУстановленияЦены
Значение: 01.01.2024 0:00:00

Запрос выводит дату, когда установлена цена на товар, наименование товара и значение цены на товар, для всех тех случаев, когда значение цены на товар было установлено с 01 января 2024 г. или позже. Запрос аналогичен запросу из 2го примера п.4.1.

РЕЗУЛЬТАТ ЗАПРОСА:



Период	Товар	Цена
10.01.2024 0:00:00	Вино "Vaziani"	4 500
10.01.2024 0:00:00	Лопти	400
10.01.2024 0:00:00	Минанкари	1 200
10.01.2024 0:00:00	Сало	160
10.01.2024 0:00:00	Масло оливковое "Grand di Oliva"	900
10.01.2024 0:00:00	Бедуинский платок	3 800
10.01.2024 0:00:00	Минеральная вода "Borjomi"	145

Для того, что бы добавить новый параметр в запрос необходимо 1) прописать его в тексте запроса (указав символ & перед названием параметра), 2) нажать на кнопку «Проверка» (после чего параметр добавится в панель параметров), 3) задать значение параметра в в панели параметров (в примере выше последовательность действий указаны стрелками).

2.6. Функции работы со строками

Основные функции работы со строками в запросах:

- Оператор «+» - конкатенация строк (т.е. добавление к одной строке другой строки).
- Оператор «Строка» - преобразует значение из любого примитивного типа в строку.
- ПОДСТРОКА(<Строка>, <НачальнаяПозиция>, <Длина>) – выбирает из строки <Строка> подстроку, подстрока начинается с позиции <НачальнаяПозиция> строки <Строка> и имеет количество символов, равное значению <Длина>.
- Лев(<Строка>, <Длина>) – выбирает из строки <Строка> подстроку, подстрока начинается с первого символа строки <Строка> и имеет количество символов, равное значению <Длина> (частный случай функции ПОДСТРОКА когда начальная позиция указывает на первый символ строки).

ПРИМЕР:

ВЫБРАТЬ
"Первые 5 символов названия товара: " + ПОДСТРОКА(Цены.Товар.Наименование,
0, 5) КАК Товар,
Лев(Строка(Цены.Период), 10) КАК Дата
ИЗ
РегистрСведений.Цены КАК Цены

Запрос выводит 2 столбца.

В первом столбце начале каждой ячейки написано "Первые 5 символов названия товара: ", после чего выводится первые 5 символов из каждого наименования товара, если наименование товара короче 5 символов, то оно выводится полностью.

Во втором столбце выводится дата установления цены на товар (без времени) - цена на товар (дата и время) преобразуется в строку и выбирается первые 10 символов строки (как раз та часть строки, где отображается дата)

РЕЗУЛЬТАТ ЗАПРОСА:

Результат1 (0,002 мс., строк: 15)	
Товар	Дата
Первые 5 символов названия товара: Папай	01.01.2020
Первые 5 символов названия товара: Папай	01.01.2021
Первые 5 символов названия товара: Вино	01.01.2021
Первые 5 символов названия товара: Масло	01.01.2021
Первые 5 символов названия товара: Бедуи	01.01.2021
Первые 5 символов названия товара: Самов	01.01.2021
Первые 5 символов названия товара: Раа	01.01.2021
Первые 5 символов названия товара: Минер	01.01.2021
Первые 5 символов названия товара: Вино	10.01.2024
Первые 5 символов названия товара: Минан	10.01.2024
Первые 5 символов названия товара: Лапти	10.01.2024
Первые 5 символов названия товара: Сало	10.01.2024
Первые 5 символов названия товара: Бедуи	10.01.2024
Первые 5 символов названия товара: Масло	10.01.2024
Первые 5 символов названия товара: Минер	10.01.2024

2.7. Функции работы с датами

Основные функции работы с датами в запросах:

- ДАТАВРЕМЯ(<Год>,<Месяц>,<День>,<Час>,<Минута>,<Секунда>) – функция, задающее значение даты и времени (см. пример 2 из п. 4.2).
- НАЧАЛОПЕРИОДА(<Дата>, <Период>) – возвращает день (со временем 00:00:00), соответствующий началу периода для указанной даты.
- КОНЕЦПЕРИОДА(<Дата>, <Период>) – возвращает день (со временем 23:59:59), соответствующий концу периода для указанной даты
- ДОБАВИТЬКДАТЕ(<Дата>, <Тип>, <Количество>) – увеличивает дату
- РАЗНОСТЬДАТ(<Дата1>, <Дата2>, <Тип>) – находит разность дат
- ГОД(<Дата>) – возвращает из указанной даты год
- КВАРТАЛ(<Дата>) – возвращает из указанной даты квартал
- МЕСЯЦ(<Дата>) – возвращает из указанной даты месяц
- ДЕНЬГОДА(<Дата>) – возвращает из указанной даты деньгода
- ДЕНЬ(<Дата>) – возвращает из указанной даты день
- НЕДЕЛЯ(<Дата>) – возвращает из указанной даты неделю
- ДЕНЬНЕДЕЛИ(<Дата>) – возвращает из указанной даты деньнедели
- ЧАС(<Дата>) – возвращает из указанной даты час
- МИНУТА(<Дата>) – возвращает из указанной даты минуту
- СЕКУНДА(<Дата>) – возвращает из указанной даты секунду

ПРИМЕР:

ВЫБРАТЬ

Заказы.Покупатель КАК Покупатель,
Заказы.Договор КАК Договор,
Заказы.Дата КАК Дата,
Квартал(Заказы.Дата) КАК Квартал

ИЗ

Документ.Заказы КАК Заказы

ГДЕ

ДАТАВРЕМЯ(2021,01,01) = НАЧАЛОПЕРИОДА(Заказы.Дата, Год)

Запрос выводит наименование покупателя, наименование договора, дату заказа, а также квартал, в котором этот заказ был осуществлен. В условии ГДЕ отбираются только реализованные в 2021 году заказы.

РЕЗУЛЬТАТ ЗАПРОСА:

Результат1 (0,001 мс., строк: 8)			
Покупатель	Договор	Дата	Квартал
ООО "ПЕРСПЕКТИВА"	Договор 1/1 от 2021	05.05.2021 12:00:01	2
ООО "ВИТА"	Договор №22	12.06.2021 9:22:58	2
ООО "ЭРА"	Основной договор	12.06.2021 12:00:02	2
ООО "ПЕРСПЕКТИВА"	Договор 1/2 от 2021	13.06.2021 23:20:54	2
ООО "ВЕГА"	Договор № 15	13.06.2021 23:21:20	2
ООО "ОРИОН"	Договор №09	13.06.2021 23:21:23	2
ООО "ВИТА"	Договор №22	30.06.2021 23:59:00	2
ООО "ОРИОН"	Договор №09	06.07.2021 12:00:00	3

2.8. УПОРЯДОЧИТЬ ПО

Оператор УПОРЯДОЧИТЬ ПО используется для сортировки результатов запроса. В операторе указываются названия столбцов, по которым необходимо провести упорядочивание, и одно из значений

- УБЫВ – упорядочивание столбца по убыванию,
- ВОЗР – упорядочивание столбца по возрастанию.

Если УБЫВ или ВОЗР не указаны, то по умолчанию используется настройка ВОЗР.

ПРИМЕР:

ВЫБРАТЬ

*

ИЗ

РегистрСведений.Цены КАК Цены

УПОРЯДОЧИТЬ ПО

Цены.Период,

Цены.Товар.Наименование УБЫВ

Запрос выводит все данные из регистра сведений «Цены», при этом данные упорядочены сначала по дате установления цены (упорядочены по возрастанию), а для тех строк, где дата одинаковая, упорядочивание идет по наименованию товара (по убыванию).

РЕЗУЛЬТАТ ЗАПРОСА:

Результат1 (0,001 мс., строк: 15)		
Период	Товар	Цена
01.01.2020 0:00:00	Папайя	290
01.01.2021 0:00:00	Самовар	6 500
01.01.2021 0:00:00	Раа	20 900
01.01.2021 0:00:00	Папайя	340
01.01.2021 0:00:00	Минеральная вода "Borjomi"	140
01.01.2021 0:00:00	Масло оливковое "Grand di Oliva"	660
01.01.2021 0:00:00	Вино "Радеда"	4 600
01.01.2021 0:00:00	Бедуинский платок	2 200
10.01.2024 0:00:00	Сало	160
10.01.2024 0:00:00	Минеральная вода "Borjomi"	145
10.01.2024 0:00:00	Минанкари	1 200
10.01.2024 0:00:00	Масло оливковое "Grand di Oliva"	900
10.01.2024 0:00:00	Лапти	400
10.01.2024 0:00:00	Вино "Vaziani"	4 500
10.01.2024 0:00:00	Бедуинский платок	3 800

2.9. ВЫБОР ... КОГДА ... ТОГДА

Оператор ВЫБОР ... КОГДА ... ТОГДА используется для реализации нескольких вариантов действий (аналогичен условному оператору ЕСЛИ в языке программирования 1С). Схематично данный оператор можно отобразить следующим образом:

ВЫБОР

КОГДА <Условие1> ТОГДА
 <Значение1>

.....
КОГДА <УсловиеN> ТОГДА
 <ЗначениеN>

Иначе
 <ЗначениеN+1>

КОНЕЦ

Если выполняется Условие1, то в указанном поле отображается Значение1 и оператор Выбрать завершается. Если условие1 не выполняется, то проверяется Условие1. Если выполняется Условие2, то в указанном поле отображается Значение2. Таким образом оператор работает либо пока не найдет условие в блоке КОГДА, которое будет выполнено (и тогда выведется размешенное в этом блоке значение и оператор ВЫБОР завершится), либо если все условия во всех блоках КОГДА не выполнены, то тогда будет подставлено в выборку то значение, которое указано в блоке ИНАЧЕ.

ПРИМЕР:

ВЫБРАТЬ

Заказ.Период КАК Период,
Заказ.Товар КАК Товар,
Заказ.Количество КАК Количество,
Заказ.Цена КАК Цена,
Заказ.Сумма КАК Сумма,
ВЫБОР

КОГДА Заказ.Сумма < 1000 ТОГДА
 "Бюджетный товар"
КОГДА Заказ.Сумма < 5000 ТОГДА
 "Товар широкого потребления"
КОГДА Заказ.Сумма < 20000 ТОГДА
 "Дорогой товар"
Иначе
 "VIP товар"

КОНЕЦ КАК ТипТовара

ИЗ

РегистрНакопления.ИсторияЗаказов КАК Заказ

ГДЕ

ГОД(Заказ.Период) = 2021 И МЕСЯЦ(Заказ.Период) = 06

Запрос выводит дату заказа, наименование товара, количество заказанного товара, действующую на тот момент времени цену, сумму за заказ и категорию товара в зависимости от потраченной на него суммы.

Если потрачено от 1 до 1000, то выводится значение "Бюджетный товар".

Если от 1001 до 5000, то выводится значение "Товар широкого потребления".

Если от 5001 до 20000, то выводится значение "Дорогой товар".

Если стоимость товара превысила 20000, то выводится значение "VIP товар".

Данные выводятся по заказам, совершенным в июне 2021 года

РЕЗУЛЬТАТ ЗАПРОСА:

Результат1 (0,001 мс., строк: 17)					
Период	Товар	Количество	Цена	Сумма	ТипТовара
12.06.2021 9:22:58	Папайя	2	340	680	Бюджетный товар
12.06.2021 9:22:58	Самовар	4	6 500	26 000	VIP товар
12.06.2021 9:22:58	Вино "Радеда"	1	4 600	4 600	Товар широкого потребления
12.06.2021 9:22:58	Бедуинский платок	2	2 200	4 400	Товар широкого потребления
12.06.2021 9:22:58	Раа	2	20 900	41 800	VIP товар
12.06.2021 12:00:02	Папайя	10	340	3 400	Товар широкого потребления
12.06.2021 12:00:02	Самовар	3	6 500	19 500	Дорогой товар
12.06.2021 12:00:02	Вино "Радеда"	2	4 600	9 200	Дорогой товар
13.06.2021 23:20:54	Вино "Радеда"	5	4 600	23 000	VIP товар
13.06.2021 23:20:54	Бедуинский платок	2	2 200	4 400	Товар широкого потребления
13.06.2021 23:20:54	Самовар	2	6 500	13 000	Дорогой товар
13.06.2021 23:20:54	Папайя	4	340	1 360	Товар широкого потребления
13.06.2021 23:21:20	Папайя	2	340	680	Бюджетный товар
13.06.2021 23:21:23	Самовар	3	6 500	19 500	Дорогой товар
30.06.2021 23:59:00	Бедуинский платок	2	2 200	4 400	Товар широкого потребления
30.06.2021 23:59:00	Папайя	1	340	340	Бюджетный товар
30.06.2021 23:59:00	Раа	2	20 900	41 800	VIP товар

2.10. Конструктор запросов

Конструктор запросов – это инструмент по работе с созданием и модификацией запросов, во многом облегчающий жизнь программиста. С помощью конструктора запросов можно ускорить написание запроса, часто более наглядно представить запрос. В большинстве случаев при работе с запросами программисты 1С пользуются конструктором запросов.

Однако в некоторых ситуациях конструктор запросов не применим, например, 1) если Вы столкнулись с уже сформированным запросом, но в нем есть какая-то ошибка, то конструктор запросов этот текст не сможет разобрать и открыть, 2) конструктором запросов не поддерживаются комментарии (если в тексте запроса есть комментарии, то воспользовавшись конструктором запросов, Вы эти комментарии удалите). Кроме того, часто требуется разбирать «чужой код» и в текстовом редакторе, понимая язык запросов, текст запроса разобрать часто гораздо легче, чем это делать в конструкторе запросов.


Далее, будут рассмотрены те разделы (вкладки) конструктора запросов, которые выполняют действия альтернативные тем, что были рассмотрены выше. Остальные разделы (вкладки) конструктора запросов будут рассмотрены в следующих темах.

Для открытия конструктора запросов необходимо нажать на кнопку «Конструктор запроса» на панели текста запросов (или нажать на белом фоне панели текста запросов и правой кнопкой мыши, раскрыв контекстное меню, и выбрать «Конструктор запросов»). Если на панели текста запросов будет указан какой-либо запрос (без синтаксических ошибок), то конструктор запросов сформирует все соответствующие этому запросу настройки



2.10.1. Вкладка «Таблицы и поля»

Во вкладке «Таблицы и поля» отражается информация о задействованных в запросе таблицах и выбранных полях. Слева (в блоке «База данных») отображается общее дерево конфигурации со всеми таблицами, в центре (в блоке «Таблицы») – те таблицы, из которых осуществляется выборка, справа (в блоке «Поля») – столбцы, которые необходимо вывести в результате запроса.




Блок «Таблицы» содержит все таблицы, которые указаны в тексте запроса в операторе ИЗ. Блок «Поля» содержит все те строки запроса, которые указаны в операторе Выбрать.


Если необходимо добавить в оператор Выбрать константное значение элементарного типа данных, либо арифметическое выражение, либо оператор ВЫБОР, либо еще что-то не из списка столбцов какой-либо таблицы, то необходимо добавить это поле самостоятельно, нажатием на кнопку с плюсиком (кнопка «Добавить») или аналогичную кнопку из контекстного меню, и в поле внизу ввести то, что требуется указать в операторе ВЫБРАТЬ.

2.10.2. Вкладка «Условия»

Во вкладке «Условия» задаются все условия, которые используются для фильтрации строк (указаны в тексте запроса в операторе ГДЕ).




На рисунке ниже приведены настройки условий для запроса из примера в п. 4.4.




2.10.3. Вкладка «Дополнительно»

В этой вкладке задаются те настройки, которые не вошли в другие вкладки. По текущей теме во вкладке «Дополнительно» используется только один из блоков «Выборка записей», в котором указываются операторы Первые N (например, Первые 3), Без повторяющихся (Разные), Разрешенные (только те данные, на чтение которых имеются права у текущей роли).



2.10.4. Вкладка «Объединения/Псевдонимы»

В блоке слева представлен список запросов (в соответствии с пакетом запросов), а справа задается псевдонимы для столбцов (полей). Эти настройки соответствуют тому, что указано после операторов «КАК» в блоке ВЫБРАТЬ.



2.10.5. Вкладка «Порядок»

На данной вкладке задается сортировка результатов запроса. Этой вкладке соответствует оператор «УПОРЯДОЧИТЬ ПО».

На рисунке ниже отображены настройки вкладки «Порядок» для примера, рассмотренного в п. 8.

